
Testing Indirect Effects for Lower level Mediation Models in SPSS 

Prepared by Ruth Mathiowetz and Daniel Bauer, 4/21/2008 

Here we provide syntax for fitting the lower-level mediation model using the MIXED 

procedure in SPSS as well as an excel calculator, SPSSEffectsCalc.xls, that performs the 

computations necessary for evaluating the average indirect and total effects.  In addition, a 

simulated data file is provided, named sim.sas7bdat, to which the lower level mediation model can 

be fit.  The population model from which the simulated data were generated has the following form: 

In the population, the fixed-effects are 0M Yd d  , a = b = .6 and c' = .2 and the variances of the 

random effects are ( ) .6MjVAR d  , ( ) .4YjVAR d  ,  ( ) ( ) .16j jVAR a VAR b   and ( ) .04jVAR c  .  

The covariance between aj and bj is 
, .113a bj j

  , and all other random effects are uncorrelated.  

These values imply that the average indirect and total effects in the population are .473 and .673, 

respectively. Last, the Level 1 residual variances are ( ) .65MijVAR e   and ( ) .45YijVAR e  .  In the 

simulated data, the number of Level 2 units (indicated by j) is N = 100, the number of observations 

within each Level 2 unit (indicated by i) is nj = 8. We recommend saving the simulated data file to a 

directory on the users computer (e.g., c:\example\) to be analyzed using the provided syntax. We 

now show the SPSS syntax for fitting the model to the data using the procedures described in Bauer, 

Preacher, and Gil (2006). 

Restructuring Data in SPSS 

The data must first be prepared for the analysis through the creation of a single dependent 

variable (Z) from the values of the mediator (M) and the distal outcome (Y).  Two selection 

variables are also created, labeled Sy and Sm, to indicate when Z represents M versus Y.  This 

rearrangement of the data is shown visually in Table 1 of Bauer, Preacher, and Gil (2006).  Here we 

show how to accomplish this rearrangement using SPSS. SAS syntax is provided in other online 

material showing how to structure the data and fit the model within SAS. The SPSS syntax for 

restructuring the data is as follows: 

*Creating Md variable to use in data restructuring.
COMPUTE Md = m . 
EXECUTE . 

*Restructuring data for multilevel analysis.
VARSTOCASES /ID = obs 
/MAKE Z FROM Md y
/INDEX = Index1(Z)
/KEEP =  m x id.

The first part of the syntax generates Md as the dependent M variable to be used to construct the 

single dependent variable (Z). Although Md is redundant with M, this redundancy allows for the 

creation of Z from Y and M (now Md) and the retention of M as a predictor of Y within Z. The 

VARSTOCASES statement begins the data restructuring, with /ID = obs creating a variable (obs) to 

identify the row at which the observations were located in the original data file. The /MAKE Z FROM 

Md y statement creates the single dependent variable (Z) by stacking the values of the dependent 

mediator (Md) and the distal outcome (Y) so each measurement appears on a separate row.  The 

statement  /INDEX = Index1(Z) creates a variable (Index1) to distinguish Y from M values. The  

/KEEP =  m x id  statement indicates which variables should be kept as fixed variables, any variables 

that should appear in each row for a given observation.  The following page includes visual 

representations of the data set with the Md variable and the restructured data set. 

First 12 observations of the data set with the Md variable: 

Restructured data: 

The following syntax creates the two selection variables labeled Sy and Sm, to indicate when Z 

represents M versus Y: 

*Creating Sy indicator variable.
RECODE 
  Index1 
  ('Md'=0)  ('y'=1)  INTO  Sy . 
VARIABLE LABELS Sy 'Sy'. 
EXECUTE . 
*Creating Sm indicator variable.
RECODE 
  Index1 
  ('Md'=1)  ('y'=0)  INTO  Sm . 
VARIABLE LABELS Sm 'Sm'. 
EXECUTE . 

The following syntax creates the product variables SMX, SYX, and SYM for the analysis: 

*Computing variables for analysis.
COMPUTE SmX = Sm * X . 
EXECUTE . 
COMPUTE SyX = Sy * X . 
EXECUTE . 
COMPUTE SyM = Sy * M . 
EXECUTE . 

The final data set should look like this: 

Revised 8/5/2015

Building the model 

The model of interest is given by this equation: 

( ) ( ) ( )ij Mj Mij j Mij ij Yj Y j Y ij j Y ij Zijij ij ij
Z d S a S X d S b S M c S X e     

The syntax for fitting this model in SPSS: 

*Multilevel model.
MIXED 

Z  WITH Sy Sm SmX SyX SyM
/FIXED = Sy Sm SmX SyX SyM  | NOINT SSTYPE(3)
/METHOD = REML
/PRINT = COVB G SOLUTION TESTCOV
/RANDOM Sy Sm SmX SyX SyM  | SUBJECT(ID) COVTYPE(UN)
/REPEATED Index1  | SUBJECT(obs*id) COVTYPE(DIAG) .

EXUCUTE . 

The statement Z  WITH Sy Sm SmX SyX SyM specifies the outcome variable (Z) and the covariates 

(Sy Sm SmX SyX SyM). The /FIXED statement identifies the fixed effects of the two selection 

variables (SY and SM), and the product variables (SMX, SYX and SYM). The NOINT option 

removes the intercept from the model and SSTYPE(3) is the default Sums of Squares test of 

significance. The /RANDOM statement identifies the coefficients that should have random effects and 

SUBJECT(ID) indicates the grouping variable. COVTYPE(UN) requests an unstructured covariance 

matrix. The /REPEATED statement is necessary to obtain a different Level 1 residual variances for Z 

when Z represents M versus Y (i.e., allowing for different residual variances for M and Y). The 

/PRINT statement requests estimates for the fixed effects (SOLUTION), the covariance matrix of the 

random effects (G), asymptotic covariance matrices for the fixed effects and covariance parameter 

estimates (COVB) (necessary for computing standard errors for the average indirect and total 

effects), and tests for the covariance estimates (TESTCOV). 

SPSS Output 

Here we will describe the SPSS Output and which elements of the output are necessary to 

calculate the indirect and total effects. We will also identify where to put those values into the excel 

calculator, SPSSEffectsCalc.xls, to generate the estimated indirect and total effects, as well as their 

95% confidence intervals. As stated previously the /REPEATED statement in our SPSS syntax allows 

for the estimation of heterogeneous σ² values for SY and SM. The estimates for the residual 

variance can be found in the Estimates of Covariance Parameters SPSS Output. 

The output indicates that the level 1 residual variance of SM is .647 and the residual variance of SY 

is .509, which are similar to the values used to generate the data ( ( ) .65MijVAR e  , ( ) .45YijVAR e  ). 

To better indicate which values in the Estimates of Fixed Effects and Covariance Matrix for Estimates of 

Fixed Effects are used in the calculations as well as where the values should be entered into the 

calculator the values have been highlighted in both the SPSS output file and the excel calculator. 

The other estimates needed for the calculations are located in the Random Effect Covariance Structure 

(G) and Covariance Matrix for Estimates of Covariance Parameters output from SPSS. As with the other 

SPSS output, the values used in the calculations and where those values should be entered into the 

calculator have been highlighted in both the Random Effect Covariance Structure (G) SPSS output and 

the excel calculator. 

The estimated sampling covariance matrix for the covariance parameter estimates are found in the 

Covariance Matrix for Estimates of Covariance Parameters  table of output. For our example, the column 

labeled UN(3,3) corresponds to the 3
rd

 row and 3
rd

 column of the Random Effect Covariance Structure

(G)  which is the variance of the random effect of SMX (VAR(aj)). Thus the column labeled 

UN(3,3) contains the sampling variance of the estimate for VAR(aj) as well as the sampling 

covariances of this estimate with all other variance/covariance parameter estimates (indicated by the 

row index). Similarly the column labeled UN(4,4) contains the sampling (co)variances for estimates 

for the variance of the random effect of SYX (VAR(c’j)) and the column labeled UN(5,5) refers to 

the estimated variance of the random effect of SYM (VAR(bj)).  What is needed for our calculations 

is the asymptotic variance of the covariance parameter COV(aj , bj), which is labeled as UN(5,3).  
The sampling variance for this estimate will then be in the UN(5,3) row and column of the 

Covariance Matrix for Estimates Covariance of Matrix Parameters SPSS output. It has been highlighted in 

both the SPSS output and the excel calculator. 

Once all the estimates from all of the SPSS output are in the spreadsheet, it  will calculate the 

formulas for the average (fixed) indirect and total effects (equations 5 and 7) and the standard errors 

(equation 9 and 10) and 95% confidence intervals (equations 11 and 12) of these average effect 

estimates.  The variances of the random indirect and total effects are also computed (equations 6 

and 8). The 95% CIs in this calculator are based on normal sampling distribution; the Monte Carlo 

(MC) method of constructing CI is not available with this calculator. 

The final calculations: 

SPSS Syntax 

*Creating Md variable to use in data restructuring.
COMPUTE Md = m . 
EXECUTE . 

*Restructuring data for multilevel analysis.
VARSTOCASES /ID = obs 
 /MAKE Z FROM Md y 
 /INDEX = Index1(Z) 
 /KEEP =  m x id 
 /NULL = KEEP. 
EXECUTE . 

*Creating Sy indicator variable.
RECODE 
  Index1 
  ('Md'=0)  ('y'=1)  INTO  Sy . 
VARIABLE LABELS Sy 'Sy'. 
EXECUTE . 

*Creating Sm indicator variable.
RECODE 
  Index1 
  ('Md'=1)  ('y'=0)  INTO  Sm . 
VARIABLE LABELS Sm 'Sm'. 
EXECUTE . 

*Computing variables for analysis.
COMPUTE SmX = Sm * X . 
EXECUTE . 
COMPUTE SyX = Sy * X . 
EXECUTE . 
COMPUTE SyM = Sy * M . 
EXECUTE . 

*Multilevel model.
MIXED 
  Z  WITH Sy Sm SmX SyX SyM 
  /FIXED = Sy Sm SmX SyX SyM  | NOINT SSTYPE(3) 
  /METHOD = REML 
  /PRINT = COVB G SOLUTION TESTCOV 
  /RANDOM Sy Sm SmX SyX SyM  | SUBJECT(ID) COVTYPE(UN) 
  /REPEATED Index1  | SUBJECT(obs*id) COVTYPE(DIAG) . 
EXECUTE . 




